

ERINNERUNGS-CD

Hermann Köberl

Ein multimediales Andenken an unsere Schulzeit, an ein gelungenes Informatikseminar oder ein besonderes Jubiläum wäre manchmal ganz nett. Wir könnten nach vielen Jahren einen Zeitsprung in die Vergangenheit machen und (hoffentlich) angenehme Erinnerungen wecken! Mit einem einfachen Visual-Basic-Programm und der entsprechenden Ausrüstung kann dieser Wunsch leicht in Erfüllung gehen.

Die Idee

Eine Namensliste erscheint und die gewünschte Person kann ausgewählt werden. Es erscheint ein charakteristisches Bild, eine Stimme mit mehr oder weniger originellen Sprüchen ertönt und ein Text mit persönlichen Daten unterstützt unsere „memories“.

Wenn eine entsprechende Ausrüstung vorhanden ist, kann auch ein kleiner Videoclip abgespielt werden.

Die erforderlichen Werkzeuge

- Eine Visual-Basic Programmierumgebung ab Version 4.
- Ein Mikrofon zum Aufnehmen der Sounddatei.
- Ein Scanner oder eine Digitalkamera für die Bilder, sowie ein einfaches Bildbearbeitungsprogramm.
- Die Videoausrüstung: Digitalvideo oder Movie-Capture-Card.
- CD-Writer.

Im ersten Moment klingt das recht anspruchsvoll, aber in einer Gruppe von Computerfreaks ist dies alles meist verfügbar. Auch unsere Schulen werden immer besser ausgestattet und eine Klassen-Erinnerungs-CD motiviert sicher alle. Es macht Spaß und man lernt viel, ohne es zu merken – der Idealzustand.

Im folgenden wird das Projekt Schritt für Schritt realisiert und ist sicherlich auch für Anfänger der „Programmierkunst“ geeignet.

Als erstes ist die Benutzeroberfläche zu entwerfen: Welche Objekte sollen wo platziert werden und welche Eigenschaften sind zu ändern. Dabei hilft uns die unten dargestellte Form mit den nummerierten Objekten aus der Werkzeugleiste.

Objektnamen, -eigenschaften und die verwendeten Ereignisse sind anschließend in einer Tabelle übersichtlich angeführt.

Mit einer ausführlichen Programmbeschreibung sowie dem kompletten, reichlich kommentierten Quellcode sollte es nicht allzu schwierig sein, das Projekt erfolgreich zu beenden.

Programmform und Objekte

(siehe Bild auf dieser Seite)

Die Tabelle

Nr.	Objekt	Name	Eigenschaften	Ereignisse/Methoden/Verwendung
0	Form	Memories	Caption, BackColor	Load
1	Listenfeld	List1	BackColor, Font..	Click, AddItem
2	Anzeige	Anzeige	BorderStyle, Stretch	
3	Bildfeld	Bild	BackColor, BorderStyle	LoadPicture
4	Textfeld	Steckbrief	BackColor, Font..,Multiline, Scrollbars	Anzeige der individuellen Texte
5	Anzeige	Wordart	BorderStyle,Picture,Stretch	Container für Wordart-Schrift
6	Bezeichnungsfeld	Film	BackColor, Font..	Click, Videoclip abspielen

Programmbeschreibung

Im Allgemeinen Programmteil sind zwei sehr nützliche API-Funktionen aus der Windows-Multimedia-Bibliothek winmm.dll deklariert:

Zum Abspielen von WAV-Dateien dient `sndPlaySound`. Dateipfad und Name der gewünschten Sounddatei sowie ein Schalter zur Steuerung der Wiedergabeart (mit oder ohne Programmunterbrechung) müssen jeweils übergeben werden. Der Rückgabewert `R` gibt an, ob die Systemanforderung erfolgreich war oder ein Fehler ausgelöst wurde.

Für die Vorführung eines AVI-Videoclips wird der Funktion `mcExecute` ein spezieller Kommandostring übergeben, welcher das Öffnen, Laden, Ausführen und Schließen der Videodatei steuert.

Dass es um eine AVI-Datei geht, wird vom Aliasname `VID` bestimmt und die Handlungnummer des Bildfeldes(Nr.3) si-

chert die Wiedergabe im gewünschten Fenster. Es ist dies wahrscheinlich die einzige Möglichkeit, ohne Playerzusatz solche AVI-Clips in ein Programm einzubauen.

Beim Formladen wird ein Ordner definiert, wo sich alle benötigten Daten befinden. Dies ist nur während der Programmentwicklung nötig, später wird diese Angabe durch einen Leerstring ersetzt!

Danach bekommt die Filmschaltfläche einen roten Hintergrund und die Gruppenliste wird ins Listenfeld geladen. Dies könnte man auch durch Einlesen einer Textdatei flexibler gestalten.



Nach Anklicken der Liste tritt das Click-Ereignis auf und der gewählte Name wird dem Bezeichner `auswahl` zugewiesen.

Dieser, bzw. die ersten 8 Buchstaben dienen später der Zuordnung der personenbezogenen Bild-, Sound- und Textdatei. Angenommen, es wurde der Name ‚HANS‘ angeklickt: Die persönlichen Dateien ergeben sich als `hans.bmp`, `hans.wav` und `hans.txt`. Das ist unbedingt bei der Herstellung dieser Dateien zu berücksichtigen. Als Grafikformate sind außer BMP auch GIF und JPEG (jpg)-Dateien möglich. Die Textdatei ist mit einem einfachen Editor, mehrzeilig und unformatiert herzustellen. Der Audiorecorder im Windows-Zubehör reicht für Tonaufnahmen vollkommen aus.

Die gewählte Grafikdatei wird nun in das Anzeigefeld (Nr. 2) geladen, die sequentielle Textdatei `name.txt` Zeile für Zeile eingelesen, mit ASCII-Codes 13 und 10 (Eingabe+Neue Zeile) ergänzt und im Textfeld (Nr.4) angezeigt. Scroll-Leisten machen bei Bedarf auch längere Texte bequem lesbar.


Abschließend übergibt die Funktion `sndPlaySound` noch den Namen der Sounddatei, welche bis zum Ende (Flag=1) abgespielt wird.

Falls ein Videoclip vorhanden ist (`memories.avi`) geschieht beim Anklicken der Schaltfläche **FILM** folgendes:

Das Bezeichnungsfeld wechselt die Farbe, die AVI-Datei wird mit dem Aliasname VID geöffnet, der Handle (interne Fensternummer) des Bildfeldes übergeben, das Bildfeld gelöscht und schließlich der Clip gestartet. Die Bildgröße der Fernsehshow muss kleiner sein, als die Ausmaße des Bildfeldes. Das Bildfeld hat die selbe Farbe wie die Form, der Rahmen ist daher unsichtbar.

Durch nochmaliges Anklicken des Bezeichnungsfeldes VIDEOCLIP wechselt die Hintergrundfarbe zurück auf rot und der MCI-Kanal wird geschlossen. Dadurch kann eine Videosequenz auch unterbrochen werden.

So sollte das fertiggestellte Projekt aussehen- mit Liste, Foto, Steckbrief und Videoclip. Die zünftigen Sprüche kann man leider nicht hören!



Das Finale

Für das fertiggestellte Projekt müssen nun noch Installationsdisketten mit dem Installationsassistenten (siehe VisualBasic-Kurs in dieser Zeitschrift) erstellt werden. Eine solche Installation ist für alle PC's erforderlich, die kein VBasic installiert haben.

Die Installationsdateien, sowie alle projektspezifischen Daten müssen nur mehr auf eine CD ‚gebrannt‘ werden und fertig ist das Erinnerungsstück.

Viel Erfolg und Spaß mit der ERINNERUNGS-CD wünscht der Autor!

Programmcode

```

Allgemeine Deklarationen
' Spielt eine WAV-Datei:
Private Declare Function sndPlaySound Lib "winmm" Alias
"sndPlaySoundA" (ByVal IpszSoundName As String, ByVal uFlags As Long)
As Long
' Spielt einen AVI-Clip in einem Bildfeld ab:
Private Declare Function mciExecute Lib "winmm.dll" (ByVal
lpstrCommand As String) As Long
Public R As Long 'Rückgabewert für Funktionsaufrufe
Public Ordner As String

Private Sub Form_Load()
'Arbeitsverzeichnis!
Ordner = "E:\PCNews\"
'Hintergrundfarbe=rot
Film.BackColor = QBColor(12)
' Namen in Liste einfügen
List1.AddItem "Eli"
List1.AddItem "Erna"
List1.AddItem "Gerti"
List1.AddItem "Hans"
List1.AddItem "Franz"
List1.AddItem "Hermann"
List1.AddItem "Harald"
List1.AddItem "Anton"
End Sub

Private Sub List1_Click()
Dim auswahl, Txt, Text As String
On Error GoTo fehler ' evt. Fehler auffangen
' angeklickter Name
auswahl = List1.List(List1.ListIndex)
' wenn nötig Namen kürzen
If Len(auswahl) > 8 Then auswahl = Left$(auswahl, 8)
' Grafik laden
Anzeige.Picture = LoadPicture(Ordner + auswahl + ".bmp")
' Textdatei
Txt = ""
Open Ordner + auswahl + ".txt" For Input As #1
' Zeile für Zeile einlesen bis EOF-Marke erricht
Do Until EOF(1)
Input #1, Text
Txt = Txt + Text + Chr$(13) + Chr$(10)
Loop
Close #1
' Texte in das Textfeld laden
Steckbrief.Text = Txt
' Sounddatei
R = sndPlaySound(Ordner + auswahl + ".wav", 1)
Exit Sub

fehler:
Resume raus
raus:
End Sub

Private Sub Film_Click()
Dim Aktion As String 'gewünschte Videofunktion
Dim AliasName As String 'Unterfunktion für Win-DLL
AliasName = "VID"
' Beim ersten Mal Anklicken
If Film.BackColor = QBColor(12) Then
' Hintergrund gelb
Film.BackColor = QBColor(14)
' Ausgesuchte Datei mit ALIAS=VID öffnen
Aktion = "open " + Ordner + "memories.avi alias " + AliasName
R = mciExecute(Aktion)
' Ausgabefenster löschen
Bild.Picture = LoadPicture
' Bild.hWnd ist die windows-"handle #" des Bildfensters
' Umwandlung in String
HwndFr$ = CStr(Bild.hWnd)
' Übergabe der Handle-Nummer für AVI-Darstellungsbereich
Aktion = "window " + AliasName + " handle " + HwndFr$
R = mciExecute(Aktion)
' Video abspielen
Aktion = "Play VID"
R = mciExecute(Aktion)
Else
' zweites Mal Klicken = unterbrechen oder Schließen
' alle Kanäle
R = mciExecute("Close all")
Film.BackColor = QBColor(12) ' Hintergrund wieder rot
' Gruppenbild wieder laden
Bild.Picture = LoadPicture(Ordner + "seminar.bmp")
End If
End Sub

```